

1. IDENTIFICATION
- 1.1 Digital-8-35-S-B
- 1.2 680 8-Bit Character Assembly Subroutines
- 1.3 November 8, 1965

2. ABSTRACT

The 680 Data Communication System 8-Bit Character Assembly Subroutines concentrate Teletype data by assembling serial-bit data into 8-bit characters and present the user with data similar to that obtained by using a 630 DCS and scanner. They also add start and stop bits to 8-bit characters and transmit them in serial-bit fashion. Full duplex lines are assumed, but the subroutines will work with half duplex if the user handles the expected echo.

3. REQUIREMENTS

3.1 Storage

The subroutines as presently coded occupy 400 octal locations plus space for internal buffering of the input and output characters and for the TTI instructions. In addition, space is used in memory page 0 and a limited number of autoindex registers are used as explained below. Within the limits described, the program can be placed anywhere in the first 4K of PDP-8 memory. The total amount of memory used including the autoindex registers and the locations in page 0 is as follows:

$$422_8 + 7n$$

where n is the number of Teletype lines to the next even multiple of eight lines if the number of lines is not already an even multiple of eight.

3.2 Subprograms and/or Subroutines

Digital-8-35-S-A
680 5-Bit Character Assembly Subroutines

for reference or in the event the user's requirements include a mixture of 5-bit and 8-bit lines.

3.3 Equipment

Minimum configuration PDP-8
680 Data Communication System hardware

3.4 Miscellaneous

3.4.1 The tag TT8BGN must be defined as the address of the start of the Teletype subroutines. It can be defined as anywhere in memory, but must be equated to the start of a PDP-8 memory page.

3.4.2 Three autoindex registers called T8AX1, T8AX2, and T8AX3 must be defined.

3.4.3 The tag TT8PG0 must be defined as the start of an area in memory page 0 where the necessary Teletype constants can be stored. An area of 17₈ registers must be reserved.

3.4.4 The tag T8OBF must be defined as the start of the area reserved for outputting the Teletype characters. It must be equal in length to the number of lines (even multiple of 8) attached to the particular set of subroutines. It can be anywhere in memory and need not start at the beginning of a memory page.

3.4.5 The tag T8OBF2 must be defined as an area equal in length to T8OBF. It is used for double-buffering the output characters to allow maximum output rate.

3.4.6 The tag T8IBF must be defined as the area for storing incoming Teletype characters and line numbers. It must be equal in length to twice the number of lines attached to the particular set of subroutines.

3.4.7 The tag T8IN must be defined as the start of the area used by the subroutines for generating the appropriate number of TTI instructions. It must be equal in length to three times the number of lines plus one register. Here again it need not be defined as the start of a memory page.

3.4.8 The tag TTCHAR must be defined as a single register in page 0.

3.4.9 In the interrupt service routine the following set or sets of instructions must appear:

```
T8S KP           /SKIP ON CLOCK FLAG
SKP             /TEST FOR NEXT INTERRUPT CAUSE
JiMP   T8DIS    /JUMP TO APPROPRIATE CLOCK INTERRUPT
              /ROUTINE
```

Because of the speed necessary for Teletype handling, the checks for clock interrupts should be the first ones in the interrupt service interrogation loop; the link bit and accumulator contents should not be saved prior to interrogation of the appropriate clock flag. If necessary for other interrupts, the link and accumulator contents should be saved only after all clock interrupts have been checked.

3.4.10 Clock IOT's

The IOT's to test the clock for the 1 state, turn the clock on, and turn the clock off must be given the correct octal definitions:

Mnemonic	Clock 1	Clock 2	Clock 3	Clock 4
T8SKP	6421	6431	6441	6451
TT8ON	6424	6434	6444	6454
TT8OFF	6422	6432	6442	6452

4. USAGE

4.2 Calling Sequence

The pseudo command T8INIT must be executed before the instruction TT8ON and also before either of the other pseudo commands T8SOF or T8SIR is executed. (See Section 4.4.1, 4.4.2 and 4.4.3 for definitions of the pseudo commands.)

4.3 Switch Settings

None

4.4 Start up and/or Entry

Three pseudo commands for using this set of subroutines are provided to the main program. They are defined as jumps to subroutines and their definitions and instructions are included in the package. These are the only commands necessary in the main program for gathering and outputting the Teletype characters. The user should note that no subroutines are included for packing or unpacking of the characters by word or even line number.

4.4.1 Teletype Initialize (T8INIT)

This command (which must be used only once in the main program) assumes that the user enters with the number of lines in the accumulator and that the register following the initialize command

contains the first line number for this type of Teletype line. This subroutine initializes all of the buffer areas, counters, and pointers, and generates the proper number of TTI instructions.

4.4.2 Skip if Output Free (T8SOF)

This instruction skips the next register in memory and transmits the character contained in register TTCHAR if the indicated output line is free. If the output line is not free, the instruction does not skip. The instruction requires that the line number over which the character is to be transmitted be in the accumulator at the time the instruction is issued. The pseudo command takes 24 μ sec minimum time, and 42 μ sec maximum time. The accumulator will be cleared when exiting from the command.

4.4.3 Skip if Input Ready (T8SIR)

This instruction skips the next location in memory and returns with the line number in the accumulator and the character placed at TTCHAR if an input character is available. If no character is available, the instruction does not skip and the accumulator is -1. Only the low order eight bits of the character at TTCHAR should be used, as additional bits representing the stop codes are also present in the character.

If no character is available, 15 μ sec are used by the pseudo instructions; if a character is available, 37.5 μ sec are used; and if the end of the storage area is reached, a maximum of 48 μ sec is used by the instruction.

4.4.4 Skip if Either Input is Ready (T8SIE)

This command is used when more than one type of Teletype line is being used. As presently coded, it assumes that two are being used, that one is an 8-bit line, and one is a 5-bit line. It skips if a character is available from either of the two types of lines being used and presents the user with the line number in the accumulator and the character in location TTCHAR. If no character is available, 37.5 μ sec are used; if a character is available, 60 μ sec are used; if the end of a buffer area is found, a maximum of 70.5 μ sec is used. As stated, the command assumes that two types of lines are being used. However, if a single type of line at two different speeds is used, the instructions, which occupy approximately 10 locations, can easily be changed to reference the correct set of subroutines.

NOTE: Since the pseudo operation references the pseudo-operation T5SIR, the user can expect an assembly error if the 8-bit subroutines are not assembled with the 5-bit subroutines. However, assuming that the user's requirements include only 8-bit lines and that the operation T8SIE would not be used, the error may either be ignored or the coding for T8SIE be deleted from the ASCII tape.

5. RESTRICTIONS

5.1 Status Active Registers

The autoindex registers defined as T8AX1, T8AX2, and T8AX3 must not be disturbed after the pseudo operation T8INIT.

6. DESCRIPTION

6.1 Discussion

These subroutines are designed to accumulate 8-bit Teletype characters to and from multiple Teletype lines connected to a PDP-8. They handle input data in serial-bit format and present the user

with character and line identification. The user presents the routines with line identification and character format data, and the routines transmit the information in serial-bit format.

Most of the PDP-8 memory is available for data buffering and for packing. A large proportion of the time however is used in buffering the Teletype lines themselves. Assuming even minor data handling is necessary before transmission (possibly to a larger computer), present estimates indicate the user cannot handle 1288-bit lines at 110 baud. Exact timing information is shown in Section 9. The user should note that the programming described involves the handling of the Teletype lines only and does not include any packing or unpacking of words, lines, or messages. The main program communicates with the Teletype subroutines via a group of pseudo commands which are described fully in Section 4.4 with examples of their usage in Section 6.2.

If the user's requirements include a mixture of 8-bit and 5-bit lines, it is necessary that the 5-bit Character Assembly Subroutines (Digital-8-35-S-A) be included with the user's programs.

6.2 Examples and/or Applications

6.2.1 To initialize the subroutines, coding similar to the following should appear in the user's program:

```
TAD      NUMLIN      /GET NUMBER OF LINES
T8INIT   /INITIALIZE SUBROUTINES
SLN      /STARTING LINE NUMBER
ION      /ENABLE INTERRUPTS
TT8ON    /TURN ON CLOCK
```

NOTE: Following these lines of coding it is necessary that the user wait 8 clock interrupts before using the pseudo-instruction T8SOF. Otherwise, the first character transmitted will be erroneous.

6.2.2 To output a character, coding similar to the following should appear:

```
TAD      CHARAC      /GET OUTPUT CHARACTER
DCA      TTCHAR      /FOR OUTPUT SUBROUTINE
TAD      LINE NO     /GET LINE NUMBER
T8SOF    /OUTPUT, SKIP IF FREE

JMP      OUTNA       /OUTPUT NOT FREE
CONTINUE /CHARACTER ACCEPTED, CONTINUE
```

6.2.3 To test for an input character available, coding similar to the following should appear:

```
T8SIR    /CHECK FOR INPUT
JMP      .-1         /WAIT FOR A CHARACTER
DCA      SAVLIN      /SAVE LINE NUMBER
TAD      TTCHAR      /GET CHARACTER INPUT
AND      THREE7      /377, CLEAR STOP BIT
```

7. METHODS

7.1 Discussion

7.1.1 Input Character Assembly

The 8-bit Character Assembly Interrupt Subroutine executes a TTI instruction for each line selected every clock interrupt. The program then scans one eighth of the character assembly words to see if a full input character has been assembled for any of the lines. If a fully assembled character is found, the program stores the character and line number in the input buffer, zeros the TTI status word, and sets the TTI character assembly word to 2000. Note that bit 1 of the character assembly word is initially set to a 1 and the rest of the character assembly word is zeros. As the character is assembled, the character assembly word is shifted one bit position to the right for the start bit and each data bit. When the link can be set to a 1 by a RTR, the character is fully assembled.

7.1.2 Output Character Handling

Initially, the pseudo operation T8SOF adds start and stop bits to the output characters and places them in the second output buffer (T8OBF2). Eventually, the interrupt subroutine transfers the characters from the second output buffer to the first output buffer (T8OBF). One eighth of the lines are scanned for output every clock interrupt. That is, for any one line one bit may be output every eight clock interrupts. The first output buffer location for a line is tested for zero or non-zero. If it is non-zero the program outputs one bit of that location and stores the remaining information back in the first output buffer. If the T8OBF location is zero, the second output buffer is tested for zero or non-zero. The second buffer location in the zero state indicates no new output. If the location is non-zero, the program outputs one bit, stores the remaining bits in the first output buffer, and zeros the second output buffer.

8. FORMAT

8.1 Input Data (T8SIR)

If the pseudo operation T8SIR skips, the input data is the following format:

8.1.1 Accumulator contains line number.

8.1.2 The lower eight bits of the register TTCHAR contain the input character.

8.3 Output Data (T8SOF)

The user presents the pseudo operation T8SOF with output characters in the following format:

8.3.1 The lower eight bits of register TTCHAR contain the output character.

8.3.2 The accumulator contains the number of the line on which the character is to be output.

9. EXECUTION TIME

9.1 Minimum

9.2 Maximum

9.3 Average

The table below indicates the percentages of machine time used for 110 baud 8-bit systems and is as accurate as is presently possible. Any additional features which may be required for the Teletype handling would add appreciably to the times shown:

TIMING TABLE

Numbers indicate the percentage of available machine time used in the average case.

No. of Lines	8-Bit 110 Baud
32	34.1 %
64	57.7 %
96	81.3 %
128	104.9 %

9.4 Timing Equations

Where n = the number of lines, the 8-bit subroutines require an average time of $8.38n+119.5$ μ sec. Clock flags (at 110 baud) occur every 1135 μ sec.

10. PROGRAM

10.3 List of Items and Pseudo Commands

10.3.1 List of Items

TT8BGN	/BEGINNING OF SUBROUTINE. MUST BE /EQUATED TO START OF A PAGE. (AREA /INCLUDES TWO PAGES).
T8AX1	/AUTOINDEX REGISTER.
T8AX2	/AUTOINDEX REGISTER.
T8AX3	/AUTOINDEX REGISTER.
TT8PG0	/START OF CONSTANT AREA IN PAGE 0. / (LENGTH 17 ₈ REGISTERS.)
T8OBF	/START OF OUTPUT BUFFER. (LENGTH = n .)
T8OBF2	/START OF SECOND OUTPUT BUFFER. / (LENGTH = n .)
T8IBF	/START OF INPUT BUFFER. (LENGTH = $2n$.)
T8IN	/START OF TTI AREA. (LENGTH = $3n+1$.)
TTCHAR	/CHARACTER AREA PAGE 0. (SINGLE REGISTER.)

10.3.2 List of Pseudo Command

Command		Times (users)		
		Minimum	Average	Maximum
T8INIT	Initialize		-	
T8SOF	Skip if output free	24	-	42
T8SIR	Skip if input ready	15	37.5	48
T8SIE	Skip if either input ready	37.5	60.0	70.5

10.4

Program Listing

CARRET	0264	OUTR	0446
CLRFLG	0101	RETCOD	0135
CODES	0126	RPOINR	0112
DELAY	0641	RUPTST	0006
ENDCOD	0137	SEVEN	0651
FIVE	1052	SKIPS	0052
INPTS	0073	SPACE	0134
KCCLT1	6402	STI OOP	0225
KCCLT2	6422	SWITCH	0600
KCCLT3	6442	SWI OOP	0616
KCCLT4	6462	SWOUTR	0631
KCCLT5	6112	SWSKIP	0636
KNOW	0041	TCFLT1	6412
KRRLT1	6406	TCFLT2	6432
KRRLT2	6426	TCFLT3	6452
KRRLT3	6446	TCFLT4	6472
KRRLT4	6466	TCFLT5	6122
KRRLT5	6116	TLSLT1	6416
KRSLT1	6404	TLSLT2	6436
KRSLT2	6424	TLSLT3	6456
KRSLT3	6444	TLSLT4	6476
KRSLT4	6464	TLSLT5	6126
KRSLT5	6114	TPCLT1	6414
KSFLT1	6401	TPCLT2	6434
KSFLT2	6421	TPCLT3	6454
KSFLT3	6441	TPCLT4	6474
KSFLT4	6461	TPCLT5	6124
KSFLT5	6111	TSFLT1	6411
K10	0107	TSFLT2	6431
K5	0110	TSFLT3	6451
K6	0111	TSFLT4	6471
K64	0652	TSFLT5	6121
LFCCODE	0136	TAITI	0241
LNFEED	0306	THO	0020
LPOINR	0046	THOIS	0027
LACT	0125	THORU	1000
LACTV	0260	THOSTR	0045
LDFX	0050	THWAIT	1022
LDFX1	0051	TRUCOD	1051
LRUPT	0253	TRULOP	1004
LUTALL	0200		
LUTCDS	0323		
LUTIN	0400		
LUTPTS	0065		

/8BIT CHARACTER ASSEMBLY SUBROUTINES
/TYPE 680 TELETYPE LINE MULTIPLEXER
/LMH 7-8-65, 8 BIT

TT8BGN=3000
TT8PG0=126
T80BF=4600
T8IBF=4200
T80BF2=5000
T8AX1=14
T8AX2=15
T8AX3=16
TTCHAR=177
T8IN=3377

TTI=6402 /TELETYPE INPUT COMMAND
TTO=6404 /TELETYPE OUTPUT COMMAND
TTCL=6411 /CLEAR LINE REGISTER
TTTL=6414 /READ LINE REGISTER
TTSL=6412 /SET LINE REGISTER, CLR AC
TTRON=6434 /TURN CLOCK ON
TTROFF=6432 /TURN CLOCK OFF
T8SKP=6431 /SKIP ON CLOCK FLAG
TTINCR=6401 /INCREMENT LINE REGISTER

*TT8PG0

0126	0000	T8INFL,	0	/INPUT READY FLAG
0127	4177	T8RFK,	T8RF-1	/TO RESET INPUT BUFFER POINTER
0130	0000	T8NL,	0	/-NUMBER OF LINES
0131	4600	T8K7,	T80BF	/K FOR 1ST OUTPUT BUFFER
0132	3021	T8K8,	T8COM	/TO ENTER COMMON ROUTINE
0133	3151	T8SOUT,	T8OUTS	/SKIP IF OUTPUT FREE
0134	3200	T8SIN,	T8INS	/SKIP IF INPUT READY
0135	3224	T8G0,	T8GDS	/INITIALIZE ROUTINE
0136	3344	T8IES,	T8SE	/SKIP IF EITHER LINE READY
0137	4600	T8OUTK,	T80BF	/POINTER TO FIRST OUTPUT BUFFER
0140	5000	T80TK2,	T80BF2	/POINTER TO 2ND OUTPUT BUFFER
0141	5000	T8K36,	T80BF2	/K FOR 2ND OUTPUT BUFFER
0142	7770	T8CNT1,	-10	/HOLD MAJOR LOOP COUNTER
0143	0000	T8CNT2,	0	/MINOR LOOP COUNTER
0144	0000	T8CNT3,	0	/COUNTER FOR INPUT BUFFER

*TT8BGN

/MULTIPLE LEVEL INTERRUPT ROUTINE

/ALLOWS MULTIPLE LEVEL INTERRUPT TO THIS ROUTINE AND UNLIMITED OTHERS

0000	2334	T8DIS,	ISZ T8LC	/LEVEL COUNTER
0001	5216		JMP T8DIS3 /2ND LEVEL INTERRUPT	
0002	3335		DCA T8SA	/SAVE ACCUMULATOR
0003	7010		RAR	/GET LINK
0004	3336		DCA T8SVLK /SAVE LINK	
0005	1000		TAD Z 0	/INTERRUPT ADDRESS
0006	3337		DCA T8SV0	/SAVE ADDRESS
0007	6414		TTRL	/RD LINE NUMBER
0010	3340		DCA T8SVLN /SAVE LINE NUMBER	
0011	6434		TTRON	/TO CLEAR FLAG ONLY
0012	6001	T8DIS2,	ION	/RE-ENABLE PROGRAM INTERRUPT
0013	1350		TAD T8K9	/STARTING LINE-1

```

3014 6413          TTSL+1          /SET LINE REGISTER, CLEAR AC
3015 5741          JMP I T8K1 /JUMP TO TTI LOOP

/2ND LEVEL INTERRUPT
3016 6434 T8DIS3, TT8ON          /CLEAR CLOCK FLAG
3017 6001          ION          /RE-ENABLE PROGRAM INTERRUPT
3020 5400          JMP I Z 0          /RETURN TO THE MAIN PROGRAM

/RETURN FROM INPUT TTI LOOP
3021 1342 T8COM, TAD T8MNC          /MINOR COUNTER, NUMBER OF LINES/8
3022 3143          DCA Z T8CNT2          /MINOR LOOP COUNTER
3023 1343          TAD T8LN          /LINE NUMBER
3024 6413          TTSL+1          /SET LINE NUMBER
3025 1537 T8COM0, TAD I Z T8OUTK          /OUTPUT WORD
3026 7450          SNA          /SOMETHING TO TRANSMIT
3027 5323          JMP T8COM8 /SEE IF WORD AVAILABLE
3030 6405          TIO+1          /INCR. LINE REGISTER AND OUTPUT
3031 3537          DCA I Z T8OUTK          /STORE WORD
3032 1414 T8COM1, TAD I Z T8AX1          /PICK UP CHARACTER ASSEMBLY WORD
3033 7110          CLL RAR          /PUT BIT 11 IN LINK
3034 7430          SZL          /CHARACTER NOT COMPLETED
3035 5301          JMP T8COM6 /STORE CHARACTER
3036 7200          CLA          /CLEAR AC FOR TAD
3037 2137 T8COM3, ISZ Z T8OUTK          /UPDATE OUTPUT ADDR
3040 2140          ISZ Z T80TK2          /UPDATE 2ND BUFFER ADDRESS
3041 2014          ISZ Z T8AX1          /UPDATE FOR NEXT INPUT LINE
3042 2014          ISZ Z T8AX1          /UPDATE FOR NEXT INPUT LINE
3043 2143          ISZ Z T8CNT2          /ARE ONE-EIGHTH OF LINES CHECKED?
3044 5225          JMP T8COM0 /CHECK NEXT LINE

3045 6414 T8COM4, TTRL          /READ LINE NUMBER
3046 3343          DCA T8LN          /SAVE LINE NUMBER
3047 2142          ISZ Z T8CNT1          /HAVE ALL LINES BEEN CHECKED
3050 5263          JMP T8COM5 /RESET AND DISMISS
3051 1344          TAD T8K2          /-10
3052 3142          DCA Z T8CNT1          /RESET MAJOR LOOP COUNTER
3053 1345          TAD T8K3          /T8IN+1
3054 3014          DCA Z T8AX1          /RESET INPUT LINE POINTER
3055 1350          TAD T8K9          /STARTING LINE-1
3056 3343          DCA T8LN          /RESET LINE NUMBER
3057 1131          TAD Z T8K7          /T8ORF
3060 3137          DCA Z T8OUTK          /RESET OUTPUT LINE POINTER
3061 1141          TAD Z T8K36          /T8ORF2
3062 3140          DCA Z T80TK2          /RESET 2ND BUFFER POINTER
3063 6002 T8COM5, IOF          /TURN OFF INTERRUPT
3064 7240          STA          /-1
3065 1334          TAD T8LC          /LEVEL COUNTER
3066 3334          DCA T8LC          /RESTORE LEVEL COUNTER
3067 1334          TAD T8LC          /LEVEL COUNTER
3070 7700          SMA CLA          /RESTORE AC ETC
3071 5212          JMP T8DIS2 /CHECK INPUT AGAIN, ETC.
3072 1340          TAD T8SVLN /LINE NUMBER
3073 6413          TTSL+1          /SET LINE REGISTER, CLR AC
3074 1336          TAD T8SVLK /PICK UP LINK
3075 7104          CLL RAL          /RESTORE LINK
3076 1335          TAD T8SA          /RESTORE ACCUMULATOR
3077 6001          ION          /ENABLE INTERRUPT

```

3100	5737		JMP I T8SV0	/RETURN TO MAIN PROGRAM
3101	7112	T8COM6,	CLI RTR	/REMOVE START CODE
3102	3415		DCA I Z T8AX2	/STORE CHARACTER
3103	6414		TTRL	/READ LINE NUMBER
3104	3415		DCA I Z T8AX2	/STORE LINE NUMBER
3105	1014		TAD Z T8AX1	/PICK UP ADDRESS POINTER
3106	1346		TAD T8K5	/-2
3107	3014		DCA Z T8AX1	/RESET POINTER
3110	3414		DCA I Z T8AX1	/ZERO STATUS AND COUNTER WORD
3111	1347		TAD T8K6	/WORD TO RESTORE ASSEMBLY WORD
3112	3414		DCA I Z T8AX1	/RESET CHARACTER ASSEMBLY WORD
3113	2126		ISZ Z T8INFL	/SET INPUT READY FLAG
3114	2144		ISZ Z T8CNT3	/HAS END OF BUFFER BEEN REACHED
3115	5237		JMP T8COM3 /UPDATE REGISTERS	
3116	1127	T8COM7,	TAD Z T8RFK	/T8RFK-1
3117	3015		DCA Z T8AX2	/RESET INPUT BUFFER ADDRESS
3120	1130		TAD T8NL	/LENGTH OF BUFFER, NUMBER OF LINES
3121	3144		DCA Z T8CNT3	/RESET LENGTH COUNTER
3122	5237		JMP T8COM3 /UPDATE REGISTERS	
3123	1540	T8COM8,	TAD I Z T8OTK2	/PICK UP DOUBLE-BUFFERED WORD
3124	7440		SZA	/NOTHING TO SEND
3125	5330		JMP .+3	/SEND NEW WORD
3126	6401		TTINCR	/INCREMENT LINE REGISTER
3127	5232		JMP T8COM1 /CONTINUE FOR INPUT	
3130	6405		TT0+1	/INCR. LINE REGISTER AND OUTPUT
3131	3537		DCA I Z T8OUTK	/STORE WORD
3132	3340		DCA IZ T8OTK2	/ZERO 2ND WORD
3133	5232		JMP T8COM1 /CONTINUE FOR INPUT	
			/CONSTANTS	
3134	7777	T8LC,	-1	/INTERRUPT LEVEL COUNTER
3135	0000	T8SA,	0	/SAVE ACCUMULATOR
3136	0000	T8SVLK,	0	/SAVE LINK
3137	0000	T8SV0,	0	/SAVE PROGRAM COUNTER
3140	0000	T8SVLN,	0	/SAVE LINE NUMBER
3141	3377	T8K1,	T8IN	/START OF TTI SERIES
3142	0000	T8MNC,	0	/MINOR LOOP COUNTER. NO OF LINES/8
3143	0000	T8LN,	0	/LINE NUMBER
3144	7770	T8K2,	-10	/TO RESET MAJOR LOOP COUNTER
3145	3400	T8K3,	T8IN+1	/TO RESET INPUT LINE POINTER
3146	7776	T8K5,	-2	/FOR SUBTRACTION
3147	2000	T8K6,	2000	/TO RESET 8-BIT ASSEMBLY WORD
3150	0000	T8K9,	0	/STARTING LINE-1
			/PSEUD OPERATIONS	
			/SKIP IF OUTPUT IS FREE AND TRANSMIT CHARACTER AT TTCHAR	
			/OTHERWISE DONT SKIP	
			/LINE NUMBER MUST BE IN AC. 24US MIN. 42US MAX.	
			T8SOF=JMS I Z T8SOUT	
			T8OUTS,	0
3151	0000			
3152	0370		AND T8K10	/177
3153	1371		TAD T8SL	/-STARTING LINE NO.
3154	1141		TAD Z T8K36	/OUTPUT BUFFER ADDR
3155	3372		DCA T8WA	/WORK AREA
3156	1772		TAD I T8WA	/OUTPUT CHARACTER
3157	7640		SZA CLA	/SKIP IF FREE
3160	5751		JMP I T8OUTS	/EXIT

```

3161 1177 TAD Z TTCHAR /PICK UP CHARACTER
3162 0373 AND T8K11 /8 BITS ONLY
3163 1374 TAD T8K12 /1400 FOR STOP CODE
3164 7104 CLL RAL /CREATE START CODE
3165 3772 DCA I T8WA /STORE CHARACTER IN TABLE
3166 2351 ISZ T8OUTS /INDEX EXIT
3167 5751 JMP I T8OUTS /EXIT
3170 0177 T8K10, 177 /FOR LINE NUMBER
3171 0000 T8SL, 0 /-STARTING LINE NUMBER
3172 0000 T8WA, 0 /WORK AREA
3173 0377 T8K11, 377 /FOR EIGHT BIT CODE
3174 1400 T8K12, 1400 /FOR STOP CODE
*TT8RGM+200
/SKIP IF CHARACTER AVAILABLE AND RETURN WITH LINE NO. IN AC
/CHAR AT TTCHAR
/OTHERWISE DO NOT SKIP. 15 US MIN, 48 US MAX, 37.5US NORMAL
/IF READY
T8SIR=JMS I Z T8SIN
3200 0000 T8INS, 0
3201 6002 IOF
3202 7240 CLA CMA /SET AC TO -1 FOR TAD
3203 1126 TAD Z T8INFL /INPUT FLAG COUNTER
3204 7510 SPA /SOMETHING AVAILABLE
3205 5221 JMP T8INON /EXIT
3206 3126 DCA Z T8INFL /RESTORE FLAG COUNTER
3207 2223 ISZ T8CNT4 /END OF BUFFER? STARTS AT -N-1
3210 5215 JMP .+5 /GET CHARACTER
3211 1130 TAD Z T8NL /-NUMBER OF LINES
3212 3223 DCA T8CNT4 /RESET COUNTER
3213 1127 TAD Z T8REF /BUFFER ADDRESS-1
3214 3016 DCA Z T8AX3 /RESET ADDRESS
3215 1416 TAD I Z T8AX3 /PICK UP CHARACTER
3216 3177 DCA Z TTCHAR /STORE CHARACTER
3217 1416 TAD I Z T8AX3 /PICK UP LINE NUMBER
3220 2200 ISZ T8INS /INDEX EXIT
3221 6001 T8INON, 10N
3222 5600 JMP I T8INS /EXIT

3223 0000 T8CNT4, 0 /-NO OF LINES
/INITIALIZATION
/ENTER WITH NUMBER OF LINES IN AC
/FORMAT T8INIT
/ 1ST LINE NO.

T8INIT=JMS I Z T8GO
3224 0000 T8COS, 0
3225 0321 AND T8K14 /377
3226 3130 DCA Z T8NL /STORE NUMBER OF LINES
3227 1130 TAD Z T8NL /NUMBER OF LINES
3230 0322 AND T8K15 /7
3231 7640 SZA CLA /MULTIPLE OF 8?
3232 1323 TAD T8K16 /10
3233 1130 TAD Z T8NL /NUMBER OF LINES
3234 0324 AND T8K17 /370
3235 7041 CIA /-TWO'S COMP NUMBER OF LINES
3236 3130 DCA Z T8NL /-N, CONSTANT
3237 1130 TAD Z T8NL /-N
3240 3144 DCA Z T8CNT3 /INPUT COUNTER

```

```

3241 1326      TAD T8K20          /T8IN-1
3242 3014      DCA Z T8AX1        /TO STORE TTI TABLE
3243 1327      TAD T8K21          /T8ORF-1
3244 3015      DCA Z T8AX2        /TO CLEAR OUTPUT AREA
3245 1343      TAD T8K37          /T8ORF2-1
3246 3016      DCA Z T8AX3        /TO CLEAR DOUBLE BUFFER
3247 1130      TAD Z T8NL          /-N
3250 3223      DCA T8CNT4        /FOR COUNTING
3251 1330      T8G02,      TAD T8K22          /TTI+INCR
3252 3414      DCA I Z T8AX1        /STORE TTI
3253 3414      DCA I Z T8AX1        /ZERO STATUS AND COUNTER WORD
3254 1331      TAD T8K23          /ASSEMBLY RESET WORD
3255 3414      DCA I Z T8AX1        /RESET ASSEMBLY WORD
3256 3415      DCA I Z T8AX2        /ZERO OUTPUT WORD
3257 3416      DCA I Z T8AX3        /CLEAR DOUBLE BUFFER
3260 2223      ISZ T8CNT4        /COUNTER
3261 5251      JMP T8G02          /DO NEXT LINE
3262 1332      TAD T8K24          /JMP I Z T8K8
3263 3414      DCA I Z T8AX1        /STORE FINAL JUMP
3264 1130      TAD Z T8NL          /-N
3265 7012      RTR              /DIVIDE BY 4
3266 7010      RAR              /DIVIDE BY 8
3267 0333      AND T8K25          /17
3270 1334      TAD T8K26          /7760, MAKE NUMBER NEGATIVE
3271 3735      DCA I T8K27        /T8MNC
3272 7240      STA              /-1
3273 3223      DCA T8CNT4        /SET COUNTER TO SKIP 1ST TIME
3274 1127      TAD Z T8RFK        /T8IRF-1
3275 3015      DCA Z T8AX2        /SET INPUT BUFFER POINTER
3276 1336      TAD T8K28          /-10
3277 3142      DCA Z T8CNT1       /MAJOR LOOP COUNTER
3300 1337      TAD T8K30          /T8IN+1
3301 3014      DCA Z T8AX1        /SET TTI POINTER
3302 1131      TAD Z T8K7         /T8ORF
3303 3137      DCA Z T8OUTK       /1ST OUTPUT BUFFER POINTER
3304 1141      TAD Z T8K36        /T8ORF2
3305 3140      DCA Z T8OUTK2      /2ND OUTPUT BUFFER POINTER
3306 7240      STA              /-1
3307 1624      TAD I T8G05        /STARTING LINE NO.
3310 3740      DCA I T8K33        /T8K9, STARTING LINE NO. -1
3311 1740      TAD I T8K33        /T8K9
3312 7040      CMA              /MAKE NEGATIVE
3313 3741      DCA I T8K34        /T8SL, -STARTING LINE NO.
3314 3126      DCA Z T8INFL       /CLEAR INPUT FLAG COUNTER
3315 7240      STA              /-1
3316 3742      DCA I T8K35        /T8LC, RESET INTERRUPT LEVEL COUNTER
3317 2224      ISZ T8G05        /INDEX EXIT
3320 5624      JMP I T8G05        /EXIT

```

```

/CONSTANTS
3321 0377      T8K14,          377          /FOR LINE NUMBER
3322 0007      T8K15,          7           /FOR EVEN MULTIPLE OF 8
3323 0010      T8K16,          10          /FOR EVEN MULTIPLE OF 8
3324 0370      T8K17,          370         /FOR EVEN MULTIPLE OF 8
3325 0001      T8K18,          1           /FOR COMPLEMENTING
3326 3376      T8K20,          T8IN-1      /FOR STORING TTI'S
3327 4577      T8K21,          T8ORF-1     /FOR OUTPUT AREA

```


```



3330 6403 T8K22,      TTI+1          /TTI + INCREMENT
3331 2000 T8K23,      2000          /ASSEMBLY RESET WORD
3332 5532 T8K24,      JMP I ≠ T8K8    /FOR FINAL JUMP
3333 0017 T8K25,      17              /FOR -N/8
3334 7760 T8K26,      7760          /FOR MAKING NEGATIVE
3335 3142 T8K27,      T8MNC         /FOR -N/8
3336 7770 T8K28,      -10          /FOR MAJOR LOOP COUNTER
3337 3400 T8K30,      T8IN+1       /FOR TTI POINTER
3340 3150 T8K33,      T8K9         /FOR STARTING LINE-1
3341 3171 T8K34,      T8SL         /-STARTING LINE NO.
3342 3134 T8K35,      T8LC         /FOR INTERRUPT LEVEL COUNTER
3343 4777 T8K37,      T8ORF2-1      /FOR DOUBLE BUFFER
/SKIP IF CHARACTER AVAILABLE FROM EITHER OF TWO TYPES OF LINES
/OTHERWISE DO NOT SKIP. 31.5US MIN, 70.5US MAX, 60US NORMAL
/IF READY
T8SIF=JMS I ≠ T8IES
3344 0000 T8SE,      0
3345 4534 T8SIR          /CHECK 8-BIT CODE
3346 5351 JMP .+3        /CHECK 5-BIT CODE
3347 2344 ISZ T8SE          /INDEX EXIT
3350 5744 JMP I T8SE      /EXIT
3351 3377 T8SIR          /CHECK 5-BIT CODE
3352 5744 JMP I T8SE      /EXIT
3353 2344 ISZ T8SE          /INDEX EXIT
3354 5744 JMP I T8SE      /EXIT
PAUSE


```

11. DIAGRAMS


11.1 Flow Charts










CHECK INPUT ON EITHER LINE
 SKIP ON INPUT WITH CHARACTER
 IN TTCHAR AND AC = LINE NUMBER







12. REFERENCES (Not Applicable)